

The Derivative of a Function

Suppose that we have a function f whose domain includes an open interval containing the value a . Then we define the *derivative of f at a* , $f'(a)$, by

$$\begin{aligned} f'(a) &\equiv \lim_{h \rightarrow 0} \frac{f(a+h) - f(a)}{h} \\ &= \lim_{x \rightarrow a} \frac{f(x) - f(a)}{x - a} \end{aligned}$$

Noting the relationship $x = a + h$ or $x - a = h$, we see that the two limits above are the same. We will use either format without remark.

It is implicit from our definition that we obtain a new function from f , namely f' , when we differentiate (take the derivative of) f . We might have a function whose value is $\sin(ax)$. Well, what is the variable here? a ? x ? For that reason, when we use Maple to evaluate derivatives we must identify the variable with respect to which we are differentiating.

```
> restart;
> f:=x->x*sin(x);
                                     f := x -> x sin(x)
> A:=diff(f(x),x);
                                     A := sin(x) + x cos(x)
> fprime:=unapply(A,x);
                                     fprime := x -> sin(x) + x cos(x)
```

You see how to differentiate a function (actually, $f(x)$ is an expression) with respect to x and may note how the output is an expression. We convert the output to a function by using **unapply**, again identifying the variable to be used. Let's see how Maple handles the process if we use the two formats in our definition.

```
> limit((f(a+h)-f(a))/h,h=0);
                                     sin(a) + a cos(a)
> limit((f(x)-f(a))/(x-a),x=a);
                                     sin(a) + cos(a) a
```


The answers are written differently, but they are obviously the same.

It will be useful to know the derivatives of some of the basic functions that we use. Without comment:

```
> diff(x^8,x);
                                     8 x^7
> diff(6,x);
                                     0
> diff(sin(x),x);
                                     cos(x)
> diff(cos(x),x);
                                     - sin(x)
> diff(exp(x),x);
                                     e^x
> diff(ln(x),x);
                                     1
                                     x
```

This exercise tells us that the derivative of the sine function is the cosine function. Let's plot both functions on the same coordinate axes. Since we know that $\sin(0) = 0$, we know that the sine graph is below the cosine graph as we start at 0. If we are looking at a color version of these notes, then the sine is the red curve. We put a lot of effort into taking limits that we can now call derivatives. The values that we were seeking were slopes. As we look at the plot which follows, assume that we are on a sled that is going to ride the sine curve. Suppose we had an instrument that would read the numerical values of the cosine (green) curve. Suppose that we had another instrument right next to the first one that would read the slope of the sled at any instant. As we are pulled by a snow-mobile we watch the two instruments and are surprised to see that their readings are the same. If the x and y scales are the same, then our slope at $x = 0$ is 1, at $x = \pi/6$ it will be $\sqrt{3}/2$, at $x = \pi/3$ it will be $1/2$ and at $x = \pi/2$ we should have a slope of 0 because we will be at the crest of our hill. Obviously, we will get negative readings as we head downhill.

```
> plot([sin(x),cos(x)],x=0..2*Pi,color=[red,green],scaling=constrained);
```



Maple Examples: We will use the value of the derivative to determine the slope of our tangent line. We will do two problems, putting the output between them, and side-by-side. Problem 1: $f(x) = \ln(x)$ and $x_0 = \sqrt{2}$.

```
> restart: with(plots): with(plottools):
```

```
> f:=x->ln(x):
```

$$f := \ln$$

```
> x0:=sqrt(2); y0:=f(x0);
```

$$x0 := \sqrt{2}$$

$$y0 := \frac{1}{2} \ln(2)$$

```
> fprime:=unapply(diff(f(x),x),x);
```

$$fprime := x \rightarrow \frac{1}{x}$$

```
> m:=fprime(x0);
```

$$m := \frac{1}{2} \sqrt{2}$$

```
> eq1:=y-y0=m*(x-x0);
```

$$eq1 := y - \frac{1}{2} \ln(2) = \frac{1}{2} \sqrt{2} (x - \sqrt{2})$$

```
> y:=solve(eq1,y);
```


$$y := \frac{1}{2} \ln(2) + \frac{1}{2} \sqrt{2} x - 1$$

```
> P1:=plot(f(x),x=1/2..4,color=red):
```


```
> P2:=plot(y,x=1/2..4,color=blue):
```

```
> P3:=line([x0,0],[x0,f(x0)],color=green):
```

```
> display(P1,P2,P3);
```



Problem 1



Problem 2

Problem 2:

```
> restart: with(plots): with(plottools):
```

```

> g:=x->x*sin(3*x);
                                 $g := c \rightarrow x \sin(3x)$ 
> x1:=7*Pi/9;  y1:=g(x1);
                                 $x1 := \frac{7}{9}\pi$ 
                                 $y1 := \frac{7}{18}\pi\sqrt{3}$ 
> gprime:=unapply(diff(g(x),x),x);
                                 $gprime := x \rightarrow \sin(3x) + 3x \cos(3x)$ 
> m1:=gprime(x1);
                                 $m1 := \frac{1}{2}\sqrt{3} + \frac{7}{6}\pi$ 
> eq2:=y-y1=m1*(x-x1);
                                 $eq2 := y - \frac{7}{18}\pi\sqrt{3} = (\frac{1}{2}\sqrt{3} + \frac{7}{6}\pi)(x - \frac{7}{9}\pi)$ 
> y:=solve(eq2,y);
                                 $y := \frac{1}{2}\sqrt{3}x + \frac{7}{6}\pi x - \frac{49}{54}\pi^2$ 
> Q1:=plot(y,x=1..Pi,color=blue):
> Q2:=plot(g(x),x=0..Pi,color=red):
> Q3:=plot(x,x=0..Pi,color=cyan):
> Q4:=plot(-x,x=0..Pi,color=cyan):
> Q5:=line([x1,0],[x1,g(x1)],color=green):
> display(Q1,Q2,Q3,Q4,Q5);

```

The output is above the problem. You will note that we included $y = x$ and $y = -x$, which was done because these functions serve as bounding functions for $\sin(3x)$.

As we scan the derivatives again, we see that the derivative of e^x is e^x . What could that possibly mean? If we took a sled ride on the curve $y = e^x$, then the instrument that reads our slope is actually reading our height! Let's try "gussying up" our function by adding a scalar multiple a to x . Now we differentiate that.

```

> diff(exp(a*x),x);
                                 $a e^{(ax)}$ 

```

In **C1M3**, *Exponential Functions*, we had an example involving radioactive decay of a material. We developed an expression that told us the amount present at any time t . Looking back, we see that A was

$$A(t) = 2.837 e^{(-.005059468471t)}$$

Using Maple, let's differentiate $A(t)$:

```

> A:=t->(2.837)*exp(-.005059468471*t);
                                 $A := t \rightarrow 2.837 e^{(-.005059468471t)}$ 
> diff(A(t),t);
                                 $-.01435371205 e^{(-.005059468471t)}$ 
> -.01435371205/2.837;
                                 $-.005059468470$ 

```

After we take the derivative of $A(t)$, we try something. We divide two coefficients and we get the third. This means that we can write $A'(t)$ as

$$\begin{aligned}
 A'(t) &= -.01435371205 e^{(-.005059468471t)} \\
 &= (-.005059468470) \cdot (2.837) e^{(-.005059468471t)} \\
 &= (-.005059468470) \cdot A(t)
 \end{aligned}$$

In other words, the *rate of change* of $A(t)$ is *proportional* to the amount present! The more that you have, the more rapidly it is disappearing.

C1M9 Problems: 1. Use Maple and the **definition** of the derivative (either form) to find $F'(a)$ for $F(x) = \sin(x) + \cos(2x)$.

In problems 2 - 4, use Maple to find the derivatives and to plot the functions with their tangent lines for the given points.

2. $f(x) = \sin(x) \ln(x)$ on $[1/4, 4]$, $x_0 = \pi/3$.
3. $g(x) = x^2 - 3x + 2 + \frac{4}{x}$ on $[1/2, 4]$, $x_1 = 2$.
3. $h(x) = e^x \sin\left(\frac{\pi e^x}{3}\right)$, on $[0, 3]$, $x_2 = 3 \ln(2)$.